| AD NUMBER |
| --- |
| **AD915784** |
| LIMITATION CHANGES |
| TO:<br>**Approved for public release; distribution is unlimited.** |
| FROM:<br>**Distribution authorized to U.S. Gov't. agencies only; Test and Evaluation; AUG 1973. Other requests shall be referred to Air Force Armament Laboratory, Attn: DLJA, Eglin AFB, FL 32542.** |
| AUTHORITY |
| **AFAL ltr, 12 Mar 1974** |

AFATL-TR-73-161

# MUTUAL AERODYNAMIC INTERFERENCE EFFECTS FOR TWO AXISYMMETRIC BODIES

AUBURN UNIVERSITY



TECHNICAL REPORT AFATL-TR-73-161
AUGUST 1973



AIR FORCE ARMAMENT LABORATORY
AIR FORCE SYSTEMS COMMAND • UNITED STATES AIR FORCE
EGLIN AIR FORCE BASE, FLORIDA

# Mutual Aerodynamic Interference Effects For Two Axisymmetric Bodies

Fred W. Martin

Charles J. Smith

Grady H. Saunders, Jr.

## FOREWORD

The work described in this report was done during the period from September 1972 through June 1973 under Contract Number F08635-71-C-0090 with the Air Force Armament Laboratory, Eglin Air Force Base, Florida. The program managers were Major William A. Miller and Captain Visi Arajs (DLJA). This work was part of a follow-on effort of the work reported as AFATL-TR-71-109, Cross-Flow Corrected Axisymemtric Solution for Multiple Body Interferences, dated August 1971.

The Computer program for this work was written by Charles J. Smith and Grady H. Saunders, Jr., Graduate Assistants, and Dr. M. A. Cutchins, Associate Professor, Aerospace Engineering, Auburn University. All experimental work under this project was accomplished by the Department of Aerospace Engineering, Auburn University, Auburn, Alabama.

This technical report has been reviewed and is approved.

FENDRICK J. SMITH, JR., Colonel, USAF
Chief, Fuzes & Mun Control Sys Division

ABSTRACT

The mutual aerodynamic interference problem for two axisymmetric bodies has been analyzed using the image system technique. In order to facilitate this analysis, it has been assumed that small perturbation solutions are valid. It is further assumed that the external stores are slender bodies and that the mutual interference can be analyzed by first assuming a cross-flow solution. The image system in the cross-flow plane consists of source-sink pairs appropriately located by using the Milne-Thomson circle theorem. The actual three-dimensional source-sink pairs are displaced from the body axis according to the cross-flow image system. Their strengths are then determined by the Rankine method. Good agreement has been found between the theoretical and experimental results.





(The reverse side of this page is blank)

TABLE OF CONTENTS

LIST OF FIGURES

LIST OF FIGURES (CONCLUDED)





(The reverse of this page is blank)

## SECTION I

## INTRODUCTION

This work is part of a continuing effort to analytically examine the flow field beneath an aircraft with external stores in an effort to produce equations for computing store separation trajectories which include the effects of mutual aerodynamic interference. The application of the cross-flow solution is an alternate approach to the time-consuming and very complex method of solving the interference problem by using a prohibitively large number of singular solutions, such as that obtained by Smith and Pierce (1) for a surface source distribution on all the bodies. Thus, the most obvious advantage of the cross-flow correction method is its relatively simple application to the case of interfering multiple dissimilar bodies.

The previous analytical solution proposed by Martin (2) in which the mutual aerodynamic interference problem was considered by using a cross-flow corrected axisymmetric solution, was re-evaluated. In that work an approximate line source distribution was assumed as the solution for the isolated axisymmetric body. This solution was then used to generate the appropriate image system as suggested by the two-dimensional cross-flow solution outlined in Reference 3. In order to obtain a better correlation with experiment and to generate a consistent technique that could be used for any axisymmetric body shape, this more general approach has been developed.

The calculation of the flow field for small disturbances requires a solution to the Prandtl-Glauert equation subject to the boundary conditions of no fluid flow into the body and vanishing perturbations at large distances from the body. Since this equation is linear, superposition of elementary solutions will allow the build-up of a complex flow which satisfies the boundary conditions. The widely used vortex-lattice and line source solutions for planar and axisymmetric flows represent two common examples of the superposition of simple solutions to form more complex flow fields. However, when systems, such as the flow fields for planar wings and axisymmetric bodies, are combined, it is not possible to simply superimpose the results for the separate systems. The interference of the flow fields requires that adjustments be made to the elementary source and/or vortex distributions in order to satisfy the body boundary conditions. The cross-flow correction discussed herein is an example of this type of adjustment for two interfering axisymmetric bodies.

This solution is based on a two-dimensional image system which utilizes the Rankine method (4) of superposition of an axial source-sink distribution to form a closed body. In the cross-flow solution, however, the sources and sinks are displaced from the axes of the bodies in such a manner as to account for the interference effects and to satisfy the boundary conditions. The derivation of the cross-flow image system (3)

utilizes the Milne-Thomson circle theorem (5) and is based on the premise that the boundary condition for two-dimensional interference flow past two circular cylinders can be satisfied by displacing the sources in each body towards each other by a specified distance.

This solution differs from that of Reference 2 in that instead of assuming that the strengths of the sources in the image system are known, as in the previous work, these values are determined so that when added to the predetermined axial source-sink distribution, the body boundary conditions are satisfied.

The new technique was applied to calculate the interference flow field for two M-117 bombs. Results of the theoretical calculations were compared with the experimental values as obtained by tests performed in the low-speed wind tunnel at the contractor facility.

The computer program used to compute the theoretical values is given as Appendix I.

## SECTION II

## THEORETICAL CONSIDERATIONS

The method of solution for analysis of the two-body aerodynamic interference problem, as developed by Martin in Reference 3, has been investigated, and a new technique for improving this method has been developed. The basis of Martin's method was the combination of the single body axisymmetric solution and the cross-flow image system [3] to approximate the interference flow field for two similar bodies of revolution in a uniform flow. It was found, however, that this technique does not satisfy the boundary conditions for the three-dimensional problem [6]. In order to correct the body boundary conditions, the axisymmetric solution was combined with a cross-flow solution consisting of a source distribution displaced from the body center line and a sink distribution of equal magnitude along the body axis as suggested by the two-dimensional cross-flow solution.

The usual assumptions of small disturbances and inviscid flow are assumed. Further, it is assumed that the flow is incompressible so that the familiar Laplace equation is the governing equation and the well-known elementary solutions, source-sink, doublet, etc., flows are available.

Since the body boundary condition for the two-body problem cannot be satisfied by simply superimposing the solutions for isolated bodies, Martin [2] developed the cross-flow image system as a possible correction. In his original work, Martin proposed a solution in terms of a continuous distributed line source which was displaced according to the two-dimensional image system. Generally, it is not possible to find a continuous line source for generating the body, and the solution is approximated by a number of finite line sources. This approach leads to geometric complexities when the image system is applied, because the line sources must be at angles corresponding to the slope of the displacement curve (Figure 1). For this reason, the line sources were replaced with a number of three-dimensional point sources and sinks. This also has the advantage of requiring half as many calculations as the solution for the same number of line sources, because the integrated effect of a line source at a field point is a function of the distance from each end of the source to the point, whereas the effect of a point source depends only on the distance between the points.

Any particular solution of Laplace's equation,

$$\frac{\partial^2 \phi}{\partial x^2} + \frac{\partial^2 \phi}{\partial y^2} + \frac{\partial^2 \phi}{\partial z^2} = 0 \qquad (1)$$




Figure 1. Image System Representation of the Two-Body Case

which is representative of an actual flow must satisfy the boundary conditions at infinity and at the surfaces of any bodies which are present in the field. A physically meaningful boundary condition is that the perturbation velocity components tend towards zero at large distances from the disturbances. This is known to be the case for source flow, as will be evident from the velocity derived from the potential function. The boundary condition along the surface of an impermeable body must express the condition of no fluid flow into the body. This condition is satisfied when the component of velocity normal to the body is zero. Equivalently, the boundary condition is satisfied when the velocity at the fluid-solid interface is tangent to the body. For an axisymmetric body, the boundary condition may be expressed mathematically as

$$\frac{dr}{dx} = \frac{u_r/U_\infty}{u_x/U_\infty+1} \tag{2}$$

where dr/dx is the slope of the body in any meridional plane, $u_r$ and $u_x$ are the perturbation velocity components in the radial and axial directions, and $U_\infty$ is the magnitude of the free-stream velocity.

For a single body at zero angle-of-attack the angular velocity component, $u_\theta$, is identically zero. In the two-body case, $u_\theta$ is not generally zero, but it does satisfy the tangential flow requirement at all points on an axisymmetric body. Therefore, equation (2) is the correct expression for the body boundary condition for the two-body case as well as for the isolated body.

## SECTION III

## SINGLE BODY SOLUTION

Assuming that any axisymmetric body can be represented by sources distributed along its center line, the total velocity potential becomes the sum of these sources plus a constant and can be expressed as

$$\phi_j = - \sum_{i}^{N} \frac{m_i}{4\pi} \frac{1}{\rho_{ij}} + \phi' \tag{3}$$

where $\rho_{ij}$ is the distance from the ith source to any point j in the flow field and $\phi'$ is a constant and represents the free-stream velocity potential.

Differentiating the summation term in equation (3) yields the perturbation velocity components which, in cylindrical coordinates r, $\theta$, x, can be expressed as

$$u_{x_j} = - \sum_{i}^{N} \frac{m_i}{4\pi} \frac{\partial}{\partial x_j} \left(\frac{1}{\rho_{ij}}\right) \tag{4}$$

$$u_{r_j} = - \sum_{i}^{N} \frac{m_i}{4\pi} \frac{\partial}{\partial r_j} \left(\frac{1}{\rho_{ij}}\right) \tag{5}$$

$$u_{\theta_j} = - \sum_{i=1}^{N} \frac{m_i}{4\pi r_j} \frac{\partial}{\partial \theta_j} \left(\frac{1}{\rho_{ij}}\right) \tag{6}$$

Due to symmetry, $u_\theta$ is zero for the case of a single body at zero angle-of-attack. As can be seen from Figure 2, $\rho_{ij}$ is not a function of $\theta$ and is given by

$$\rho_{ij} = [(x_j - x_i)^2 + r_j^2]^{\frac{1}{2}} \tag{7}$$

If a new variable $z_i$, defined such that it is proportional to the source strength, i.e.,




Figure 2. Geometric Relations for the Single Body Case.

$$z_i = \frac{m_i}{4\pi U_\infty} \tag{8}$$

is introduced and equations (4) and (5) are used, the velocity components can be written in the forms,

$$\frac{u_{x_j}}{U_\infty} = -\sum_i^N z_i \frac{x_i - x_j}{\rho_{ij}^3} \tag{9}$$

$$\frac{u_{r_j}}{U_\infty} = +\sum_i^N z_i \frac{r_j}{\rho_{ij}^3} \tag{10}$$

Equations (9) and (10) can be applied at any non-singular point in the flow field to determine the velocity ratios. Referring to equation (2), it can be seen that the body boundary condition in terms of the velocity ratios is specified by

$$\frac{dr_j}{dx_j} = \frac{\sum_i^N z_i \frac{r_j}{\rho_{ij}^3}}{1 - \sum_i^N z_i \frac{x_i - x_j}{\rho_{ij}^3}} \tag{11}$$

By multiplying both sides of equation (11) by the denominator on the right-hand side of that equation and simplifying the resulting equation, one may obtain the result,

$$\sum_i^N z_i \frac{(x_i - x_j)\frac{dr_j}{dx_j} + r_j}{\rho_{ij}^3} = \frac{dr_j}{dx_j} \qquad (j=1,2,3,\ldots,N) \tag{12}$$

Equation (12) forms a set of n simultaneous linear equations for determining the required source strengths in terms of the known geometry. If the coefficients of $z_i$ in equation (12) are defined to be $c_{ij}$, the equation may be written in the form

$$\sum_i^N z_i c_{ij} = \frac{dr_j}{dx_j} \qquad (j=1,2,3,\ldots,N) \tag{13}$$

Equation (13) may be written in the equivalent matrix form,

$$[C](Z) = (dr/dx) \tag{14}$$

Solving for (Z) by matrix inversion, one obtains the result

$$(Z) = [C]^{-1}(dr/dx) \tag{15}$$

The array of strengths as calculated by this method satisfies the body boundary condition at the control points for any isolated axisymmetric body at zero angle-of-attack.

The predicted pressure distribution for an M-117 bomb is compared with experiment in Figure 3 and shows good agreement.




Figure 3. Comparison of the Theoretical and Experimental Pressure Distribution on a Single Body (M-117 Bomb).

## SECTION IV

## SOLUTION FOR TWO SIMILAR BODIES

Consider the case of two identical axisymmetric bodies aligned such that one is directly above or to the side of the other. According to Reference 2, the image system for this case is formed by displacing each source towards its corresponding source in the other body, as in Figure 4. The distances which the sources are displaced are derived from repeated applications of the Milne-Thomson circle theorem [3] and are given by

$$\delta_i = \frac{y_o}{2} \{1-[1-(2r_i/y_o)^2]^{\frac{1}{2}}\} \tag{16}$$

where $y_o$ is the distance between body center lines and $r_i$ is the radius of the body at the ith source. Since the bodies are identical, the sources at corresponding axial locations in each body are of the same strength.

In the two-dimensional case, the image system consists of the displaced sources with corresponding sinks of equal strength at the center line. Since the strengths of the image pairs at any cross-section are the same as that of the original source, from the single body solution, the sink cancels this source. Hence, the two-dimensional system yields only the displaced line of sources as discussed above. In the three-dimensional case, however, this simple system does not produce a satisfactory result.

In the three-dimensional case, there is a significant component of flow in the axial direction in addition to the radial flow. It is reasoned, however, that an equivalent line source can be assumed at the body center line which contributes only to the radial flow. This is purely a fictitious flow but suffices to justify an equivalent two-dimensional flow field which would generate the source-sink pair described in the image system above. Note that the source displacement distance, $\delta$, depends only on the geometric relations and is independent of the two-dimensional source strength.

It follows that an appropriate system for the three-dimensional case can be generated by retaining the sources along the center line of the two bodies (isolated solution) and adding an image system consisting of source-sink pairs in each body which are located according to the two-dimensional displacement distances.




Figure 4. Geometric Relations for the Two-Body Case.

The total velocity potential for this two-body system can now be expressed as before and, for a body formed by N sources, the potential at a point is

$$\phi_j = -\sum_{i=1}^{N} \frac{m_i}{4\pi}\left(\frac{1}{\rho'_{1_{ij}}} + \frac{1}{\rho'_{2_{ij}}}\right) - \sum_{i}^{N} \frac{m'_i}{4\pi}\left[\left(\frac{1}{\rho_{1_{ij}}} + \frac{1}{\rho_{2_{ij}}}\right) - \left(\frac{1}{\rho'_{i_{ij}}} + \frac{1}{\rho'_{2_{ij}}}\right)\right] \quad (17)$$

where $m_i$ refers to the single body source strength and $m'_i$ is the strength of the added source and sink. $\rho_{1_{ij}}$, $\rho_{2_{ij}}$, $\rho'_{1_{ij}}$, and $\rho'_{2_{ij}}$ are as illustrated in Figure 4 and may be expressed as

$$\rho_{1_{ij}} = [(\delta_i - r_j \sin\theta)^2 + (x_j - x_i)^2 + (r_j \cos\theta)^2]^{\frac{1}{2}} \quad (18)$$

$$\rho_{2_{ij}} = [(y_o - \delta_i - r_j \sin\theta)^2 + (x_j - x_i)^2 + (r_j \cos\theta)^2]^{\frac{1}{2}} \quad (19)$$

$$\rho'_{1_{ij}} = [(x_j - x_i)^2 + r_j^2]^{\frac{1}{2}} \quad (20)$$

$$\rho'_{2_{ij}} = [(y_o - r_j \sin\theta)^2 + (x_j - x_i)^2 + (r_j \cos\theta)^2]^{\frac{1}{2}} \quad (21)$$

Using equations (4), (5) and (6) for the velocity components and equation (8) for $z_i$, one may obtain the velocity ratios

$$\frac{u_{x_i}}{U_\infty} = -\sum_{i=1}^{N} [z_i a'_{ij} + z'_i (a_{ij} - a'_{ij})] \quad (22)$$

$$\frac{u_{r_i}}{U_\infty} = -\sum_{i=1}^{N} [z_i b'_{ij} + z'_i (b_{ij} - b'_{ij})] \quad (23)$$

$$\frac{u_{\theta_j}}{U_\infty} = -\frac{1}{r_j} \sum_{i=1}^{N} z_i d'_{ij} + z'_i (d_{ij} - d'_{ij})] \quad (24)$$

where

$$a_{ij} = \frac{\partial}{\partial x_j}\left(\frac{1}{\rho_{1_{ij}}} + \frac{1}{\rho_{2_{ij}}}\right) \quad (25)$$

$$b_{ij} = \frac{\partial}{\partial r_j}\left(\frac{1}{\rho_{1_{ij}}} + \frac{1}{\rho_{2_{ij}}}\right) \tag{26}$$

$$d_{ij} = \frac{\partial}{\partial \theta}\left(\frac{1}{\rho_{1_{ij}}} + \frac{1}{\rho_{2_{ij}}}\right) \tag{27}$$

The terms $a'_{ij}$, $b'_{ij}$, and $d'_{ij}$ are formed from equations (25), (26), and (27) with the values for $\rho_{ijk}$ replaced with the primed quantities (see Figure 4 and equations (18) to (21)). The unprimed values of z correspond to the single body source strengths, and the primed values represent the image system.

The source distribution to correct for the interference is developed by application of the requirement of tangential flow at the body. Using equations (22) and (23) for the velocity components and equation (2) for the boundary condition, one may obtain the result

$$\frac{dr_j}{dx_j} = \sum_i^N \{z_i(a'_{ij}\frac{dr_j}{dx_j} - b'_{ij}) + z'_i[(a_{ij}\frac{dr_j}{dx_j} - b_{ij}) - (a'_{ij}\frac{dr_j}{dx_j} - b'_{ij})]\} \tag{28}$$

Equation (28) represents a set of N simultaneous equations for determining the source distribution required to account for the interference between two similar bodies. By defining the coefficients of $z_i$ and $z'_i$ as $e_{ij}$ and $f_{ij}$, respectively, equation (28) can be written in the equivalent matrix form

$$(dr/dx) = [e](z) + [f](z') \tag{29}$$

Solving this equation for (z') one obtains

$$(z') = [f]^{-1}\{(dr/dx) - [e](z)\} \tag{30}$$

The source distribution given by equation (30) will satisfy the body boundary conditions at N control points and give good agreement at all other points around the body. It was found that the best location for the control points lies along the intersection of one body with the plane which passes through the center lines of both bodies; i.e., from Figure 4, the 90-degree meridional plane. The predicted slopes at other meridional planes are shown on Figure 5 for the M-117 bomb body shape.

The predicted pressure distribution and experimental points are shown in Figures 6 to 11 for meridional angles between 90° and -90°. These

distributions were calculated using a 30-point solution for the axisymmetric body and for the cross-flow correction.




Figure 5. Comparison of Body Slope with Computed Streamlines at Meridional Angles, $\theta = -90°, 0°, 90°$.




Figure 6. Comparison of Theoretical and Experimental Pressure Distribution for the Two-Body Case, $y_o$ = 1.042 Dia., θ = 90°.




Figure 7. Comparison of Theoretical and Experimental Pressure Distribution for the Two-Body Case, $y_o$ = 1.042 Dia., $\theta$ = 60°.




Figure 8. Comparison of Theoretical and Experimental Pressure Distribution for the Two-Body Case, yo = 1.042 Dia., $\theta = 30°$.




Figure 9. Comparison of Theoretical and Experimental Pressure Distribution for the Two-Body Case, $y_o$ = 1.042 Dia., $\theta$ = 0°.




Figure 10. Comparison of Theoretical and Experimental Pressure Distribution for the Two-Body Case, $y_o$ = 1.042 dia., $\theta$ = - 30°.




Figure 11. Comparison of Theoretical and Experimental Pressure Distribution for the Two-Body Case, $y_o$ = 1.042 Dia., $\theta$ = - 90°.

## SECTION V

## SOLUTION FOR TWO DISSIMILAR BODIES

The two-body solution of the previous section can be extended to include the case of dissimilar axially misaligned bodies. Again, and in the solution for two similar bodies, it will be assumed that the isolated body solution has an equivalent two-dimensional source distribution which causes the radial flow outward from the body centerline. Also, it should be noted that this radial flow occurs both forward and aft of the body as well as along the body. Thus, in order to generate the cross-flow image system, it is assumed, conceptually, that the flow field forward and aft of an isolated body is spread by equivalent sources distributed along the extension of the body center line. The image system in body 2 at cross-flow station a (see Figure 12) can be generated using the Milne-Thomson circle theorem for source flow over a cylinder and is illustrated as Figure 13(a) where $\delta_2$ is given by

$$\delta_2 = \frac{r_2^2}{y_o} \tag{31}$$

The complete image system at station b can be generated by repeated application of the Milne-Thomson circle theorem and consists of a line of sources and sinks within both bodies. The complete development is given in Reference 3. It has been found, however, that one source-sink for each body with the source displacement distances given by

$$\delta_2 = \cfrac{r_2^2}{y_o - \cfrac{r_1^2}{y_o - \cfrac{r_2^2}{y_o - \cfrac{r_1^2}{y_o}}}} \tag{32}$$




Figure 12. Geometric Relations for Two Misaligned Bodies.




Figure 13. Schematic of Cross-Section for Dissimilar Bodies.

and

$$\delta_1 = \cfrac{r_1^2}{y_o - \cfrac{r_2^2}{y_o - \cfrac{r_1^2}{y_o - \cfrac{r_2^2}{y_o}}}} \qquad (33)$$

yields good results as will be shown later. Also note that equations (32) and (34) reduce to the truncated series representing equation (16) for $\delta$ in the solution for two similar bodies where $r_1 = r_2$.

Because of the asymmetry of this case, the strength and the displacements of the source-sink pairs are different for the two bodies. Thus, if there are N image pairs in each body (corresponds to N sources for the isolated body solution), then there will be a set of 2N simultaneous equations to be solved.

The velocity potential function for this case can be expressed as

$$\frac{\phi_j}{U_\infty} = -\sum_{i=1}^{N} z_{1_i} \left(\frac{1}{\rho'_{1_{ij}}}\right) - \sum_{i=1}^{N} z'_{1_i} \left(\frac{1}{\rho_{1_{ij}}} - \frac{1}{\rho'_{1_{ij}}}\right)$$

$$- \sum_{i=1}^{N} z_{2_i} \left(\frac{1}{\rho'_{2_{ij}}}\right) - \sum_{i=1}^{N} z'_{2_i} \left(\frac{1}{\rho_{2_{ij}}} - \frac{1}{\rho'_{2_{ij}}}\right) + \frac{\varphi'}{U_\infty} \qquad (34)$$

where the subscripts 1 and 2 correspond to bodies 1 and 2, respectively, as shown on Figure 12. The other terms are consistent with the previous notation and as defined by Figure 12.

Performing the appropriate differentiations on equation (34) in order to obtain the velocity components and using equation (2) for the boundary condition generates the set of 2N simultaneous equations which can be written in matrix form as equation (30). The z' values in this case are N values of $z'_1$ and N values of $z'_2$ corresponding to the image source-sink pairs in bodies 1 and 2, respectively.

The N control points on body 1 were taken along the top of the body at the intersection of the meridional plane passing through both body center lines. The other N control points were taken at the corresponding lower surface of body 2.

The predicted pressure distribution and experimental results for body 1 with body 1 displaced two diameters downstream from body 2 are shown on Figure 14.




Figure 14. Comparison of Theoretical and Experimental Pressure Distribution for the Solution for Dissimilar Bodies, $x_o$ = 2.0 Dia. and $y_o$ = 1.042 Dia.

## SECTION VI

## CONCLUSIONS

This method led to good agremeent between the experimental and theoretical pressure distributions. Also, the slope of the body streamline calculated using this method and the slope of the actual body were in good agreement. Thus, this method can be considered a valid solution for the mutual aerodynamic interference problem for two axisymmetric bodies at zero angle-of-attack.



(The reverse of this page is blank)

# APPENDIX I

## FORTRAN COMPUTER PROGRAM FOR

## 2-BODY INTERFERENCE

The program which follows is the FORTRAN computer program used to calculate the flow field for two parallel bodies in close proximity to each other by the source-sink image pairs approach. The program was written for the IBM 360 system and compiled on the FORTRAN G Compiler. All calculations were performed in single precision arithmetic.

```
C
C          TWO SIMILAR BODIES
C
      DIMENSION RAD(31),X(31),SLOPE(31),XA(30),XB(30),DRDXA(30),
     -ZA(30)  ,DRDXB(30),DELA(30),DELB(30)
      DIMENSION RA(30), RB(30), ZB(30)
      COMMON /TWO/  XA,XB,RA,DRDXA,DRDXB,DELA,DELB,ZA
      EQUIVALENCE (RA(1),RB(1),RAD(1)), (ZA(1),ZB(1))
C
      NS=30
      DX=0.20
      XO=0.0
      CALL GEOM(RAD,X,DX,NS,SLOPE)
  400 CONTINUE
      IF( XO .NE. 0.0 ) GO TO 18
C
C          SINGLE BODY SOURCE STRENGTHS
C
      YO=10000.0
      GO TO 17
C
C          TWO-BODY SOLUTION
C
   18 CONTINUE
      YO=1.042
   17 DO 200 I=1,NS
      XA(I)=X(I)+XO
      XB(I)=X(I)
      DRDXA(I)=SLOPE(I)
      DRDXB(I)=-SLOPE(I)
  200 CONTINUE
      IF(YO.GE.10000.0) GO TO 16
      DO 150 I=1,NS
      M=XA(I)/DX
      N=(XB(I)-XO)/DX
      IF(M.GT.NS) GO TO 156
      DELA(I)=RA(I)**2/(YO-(RB(M)**2)/YO)
  156 CONTINUE
      IF(XA(I).GE.XB(NS)) DELA(I)=RA(I)**2/YO
      IF(N.LE.0) GO TO 155
      DELB(I)=RB(I)**2/(YO-(RA(N)**2)/YO)
  155 CONTINUE
      IF(XB(I).LT.XO) DELB(I)=RB(I)**2/YO
  150 CONTINUE
C
      WRITE(6,110) XO,YO
  110 FORMAT(1H1,'  TWO-BODY SOLUTION'/' XO = ',F10.5,' YO = ',
     -F10.5//'  I      DELA(I)    DELB(I)'//)
```

```
      WRITE(6,111)(I,DELA(I),DELB(I),I=1,NS)
  111 FORMAT(1X,I2,2F12.5/)
   16 CALL CROSS2(NS,XO,YO)
      IF(YO.GE.10000.0) GO TO 18
      IF( XO .NE.0.0 ) GO TO 300
      XO=2.0
      GO TO 400
  300 STOP
      END
```

```
      SUBROUTINE CROSS2(NS,XO,YO)
C
      DIMENSION ZCO(60,60), E(60), DELA(30),DELB(30), X(30),
     -    R(30),XA(30), XB(30),DRDXA(30),DRDXB(30),ZA(30)
     -  ,UX(30),UR(30),BC(30),CP(30)
      DIMENSION PC(30,30), RA(30), RB(30),ZB(30),ZAP(30),ZBP(30)
      DIMENSION UT(30)
      COMMON /TWO/  XA,XB,RA,DRDXA,DRDXB,DELA,DELB,ZA
      EQUIVALENCE (PC(1,1),ZCO(1,1)),
     -            (RA(1),RB(1)),(ZA(1),ZB(1)),
     -            (ZAP(1),E(1)), (ZBP(1),E(31))
      SIND(PPP)= SIN(PPP/57.29578)
      COSD(PPP)= COS(PPP/57.29578)
C
C
C        EVALUATION ALONG BODY A
C
      DO 301 J=1,NS
      X(J)=XA(J)
      R(J)=RA(J)
      E(J)=DRDXA(J)
  301 CONTINUE
C
      DO 100 J=1,NS
      DO 100 I=1,NS
C
      PA1= SQRT((X(J)-XA(I))**2+R(J)**2)
   11 AX=(X(J)-XA(I))/PA1**3
   12 AR=R(J)/PA1**3
C
      IF(YO.LT.10000.0) GO TO 10
      PC(J,I)=AR-AX*DRDXA(J)
      GO TO 100
C
   10 PA2= SQRT((X(J)-XA(I))**2+(DELA(I)-R(J))**2)
      PB1= SQRT((X(J)-XB(I))**2+(R(J)-YO)**2)
      PB2= SQRT((X(J)-XB(I))**2+(R(J)+DELB(I)-YO)**2)
C
      BX=(X(J)-XA(I))/PA2**3-(X(J)-XA(I))/PA1**3
      CX=(X(J)-XB(I))/PB1**3
      DX=(X(J)-XB(I))/PB2**3-(X(J)-XB(I))/PB1**3
C
      BR=(R(J)-DELA(I))/PA2**3-R(J)/PA1**3
      CR=( (J)-YO)/PB1**3
      DR=(R(J)+DELB(I)-YO)/PB2**3-(R(J)-YO)/PB1**3
C
      AA=AR-AX*DRDXA(J)
C
```

```
      CA=CR-CX*DRDXA(J)
C
      ZCO(J,I)=BR-BX*DRDXA(J)
      ZCO(J,I+NS)=DR-DX*DRDXA(J)
      E(J)        = E(J)-ZA(I)*AA-ZB(I)*CA
C
  100 CONTINUE
C
      IF(YO.LT.10000.0) GO TO 14
      CALL SIMQ (PC,DRDXA,NS,KS)
      DO 15 I=1,NS
   15 ZA(I)=DRDXA(I)
      RETURN
C
   14 CONTINUE
C
C
C        EVALUATION ALONG BODY B
C
      DO 302 J=1,NS
      X(J)=XB(J)
      R(J)=YO-RB(J)
      E(J+NS)=DRDXB(J)
  302 CONTINUE
C
      DO 200 J=1,NS
      DO 200 I=1,NS
C
      PA1= SQRT((X(J)-XA(I))**2+R(J)**2)
      PA2= SQRT((X(J)-XA(I))**2+(DELA(I)-R(J))**2)
      PB1= SQRT((X(J)-XB(I))**2+(R(J)-YO)**2)
      PB2= SQRT((X(J)-XB(I))**2+(R(J)+DELB(I)-YO)**2)
C
      AX=(X(J)-XA(I))/PA1**3
      BX=(X(J)-XA(I))/PA2**3-(X(J)-XA(I))/PA1**3
      CX=(X(J)-XB(I))/PB1**3
      DX=(X(J)-XB(I))/PB2**3-(X(J)-XB(I))/PB1**3
C
      AR=R(J)/PA1**3
      BR=(R(J)-DELA(I))/PA2**3-R(J)/PA1**3
      CR=(R(J)-YO)/PB1**3
      DR=(R(J)+DELB(I)-YO)/PB2**3-(R(J)-YO)/PB1**3
C
C
      AB=AR-AX*DRDXB(J)
      CB=CR-CX*DRDXB(J)
C
      ZCO(J+NS,I)    =BR-BX*DRDXB(J)
```

```
      ZCC(J+NS,I+NS)=DR-DX*DRDXB(J)
      E(J+NS)=      E(J+NS)-ZA(I)*AB-ZB(I)*CB
C
  200 CONTINUE
      NS2=2*NS
C
      CALL SIMQ (ZCC,E,NS2,KS)
      IF(QS.EQ.1.0) GO TO 2200
      GO TO 300
 2200 WRITE(6,400)
  400 FORMAT(1H1,'SOLUTION FOR SOURCE STRENGHTS IS SINGULAR',/)
      RETURN
  300 WRITE(6,500)
  500 FORMAT(1H1,//,1X,'NON-DIMENSIONAL SOURCE STRENGHTS',//,4X,
     -'I',7X,'Z(I)',//)
      WRITE(6,501) (I,E(I),I=1,NS2)
  501 FORMAT(1X,I4,E16.8)
C  ZAP(I) ARE THE NON-DIMENSIONAL STRENGTHS OF THE SOURCE-SINK
C PAIRS IN BODY A  THAT ARE SOLVED FOR IN SUBROUTINE SIMQ.
C  ZBP(I) ARE THE NON-DIMENSIONAL STRENGTHS OF THE SOURCE-SINK
C PAIRS IN BODY B  THAT ARE SOLVED FOR IN SUBROUTINE SIMQ.
C
C         EVALUATE CP AND BC ALONG BODY A
C
      DO 303 J=1,NS
      UX(J)=0.0
      UR(J)=0.0
      BC(J)=0.0
      CP(J)=0.0
      X(J)=XA(J)
      R(J)=RA(J)
  303 CONTINUE
C
      DO 4400 J=1,NS
      DO 405 I=1,NS
      PA1= SQRT((X(J)-XA(I))**2+R(J)**2)
      PA2= SQRT((X(J)-XA(I))**2+(DELA(I)-R(J))**2)
      PB1= SQRT((X(J)-XB(I))**2+(R(J)-YO)**2)
      PB2= SQRT((X(J)-XB(I))**2+(R(J)+DELB(I)-YO)**2)
C
      AX=(X(J)-XA(I))/PA1**3
      BX=(X(J)-XA(I))/PA2**3-(X(J)-XA(I))/PA1**3
      CX=(X(J)-XB(I))/PB1**3
      DX=(X(J)-XB(I))/PB2**3-(X(J)-XB(I))/PB1**3
C
      AR=R(J)/PA1**3
      BR=(R(J)-DELA(I))/PA2**3-R(J)/PA1**3
      CR=(R(J)-YO)/PB1**3
```

```
      DR=(R(J)+DELB(I)-YO)/PB2**3-(R(J)-YO)/PB1**3
C
C
C
      UX(J)=UX(J)+ZA(I)*AX+ZAP(I)*BX+ZB(I)*CX+ZBP(I)*DX
      UR(J)=UR(J)+ZA(I)*AR+ZAP(I)*BR+ZB(I)*CR+ZBP(I)*DR
  405 CONTINUE
C
      BC(J)=UR(J)/(1.0+UX(J))
      CP(J)=-2.0*UX(J)-UX(J)**2-UR(J)**2
C
 4400 CONTINUE
      WRITE(6,600) XO,YO
  600 FORMAT(1H1,' BODY A   XO = ',F10.5,'  YO = ',F10.5/
     -'  I          XA             RA             DRDXA          BC
     -UX            UR             CP             ZA            ZAP'//)
2     WRITE(6,700)(I,XA(I),RA(I),DRDXA(I),BC(I),UX(I),UR(I)
     - , CP(I), ZA(I), ZAP(I), I=1,NS )
  700 FORMAT(1X,I2,9F13.5/)
C
C         EVALUATE CP AND BC ALONG BODY B
C
      DO 304 J=1,NS
      UX(J)=0.0
      UR(J)=0.0
      BC(J)=0.0
      CP(J)=0.0
      X(J)=XB(J)
      R(J)=YO-RB(J)
  304 CONTINUE
C
      DO 5500 J=1,NS
      DO 505 I=1,NS
      PA1= SQRT((X(J)-XA(I))**2+R(J)**2)
      PA2= SQRT((X(J)-XA(I))**2+(DELA(I)-R(J))**2)
      PB1= SQRT((X(J)-XB(I))**2+(R(J)-YO)**2)
      PB2= SQRT((X(J)-XB(I))**2+(R(J)+DELB(I)-YO)**2)
C
      AX=(X(J)-XA(I))/PA1**3
      BX=(X(J)-XA(I))/PA2**3-(X(J)-XA(I))/PA1**3
      CX=(X(J)-XB(I))/PB1**3
      DX=(X(J)-XB(I))/PB2**3-(X(J)-XB(I))/PB1**3
C
      AR=R(J)/PA1**3
      BR=(R(J)-DELA(I))/PA2**3-R(J)/PA1**3
      CR=(R(J)-YO)/PB1**3
      DR=(R(J)+DELB(I)-YO)/PB2**3-(R(J)-YO)/PB1**3
C
```

```
C
      UX(J)=UX(J)+ZA(I)*AX+ZAP(I)*BX+ZB(I)*CX+ZBP(I)*DX
      UR(J)=UR(J)+ZA(I)*AR+ZAP(I)*BR+ZB(I)*CR+ZBP(I)*DR
C
  505 CONTINUE
C
      BC(J)=UR(J)/(1.0+UX(J))
      CP(J)=-2.0*UX(J)-UX(J)**2-UR(J)**2
C
 5500 CONTINUE
      WRITE(6,800) XO,YO
  800 FORMAT(1H1,' BODY B   XO = ',F10.5,'  YO = ',F10.5/
     -'  I          XB           RB           DRDXB          BC
     -UX           UR           CP           ZB           ZBP'//)
      WRITE(6,700)(I,XB(I),RB(I),DRDXB(I),BC(I),UX(I),UR(I)
     - , CP(I), ZB(I), ZBP(I), I=1,NS )
C
C
C        EVALUATION OF UX, UR, UT, CP, AND BC AROUND BODY A
C
      T=90.0
   24 DO 20 J=1,NS
      UX(J)=0.00
      UR(J)=0.00
      UT(J)=0.00
      BC(J)=0.00
      CP(J)=0.00
      X(J)=XA(J)
      R(J)=RA(J)
   20 CONTINUE
C
      DO 21 J=1,NS
      DO 22 I=1,NS
C
      PXA= X(J)-XA(I)
      PXB= X(J)-XB(I)
      PZ= COSD(T)*R(J)
C
      PA1 = SQRT( PXA**2+PZ**2+( SIND(T)*R(J)          )**2 )
      PA2 = SQRT( PXA**2+PZ**2+( SIND(T)*R(J)-DELA(I)  )**2 )
      PB1 = SQRT( PXB**2+PZ**2+(-SIND(T)*R(J)      +YO)**2 )
      PB2 = SQRT( PXB**2+PZ**2+(-SIND(T)*R(J)-DELB(I)+YO)**2 )
C
      AX=PXA/PA1**3
      BX=PXA/PA2**3 - AX
      CX=PXB/PB1**3
      DX=PXB/PB2**3 - CX
C
```

```
      AR=                              R(J)/PA1**3
      BR=          (R(J)-SIND(T)*DELA(I))/PA2**3 - AR
      CR=               -(SIND(T)*YO-R(J))/PB1**3
      DR=-(SIND(T)*(YO-DELB(I))-R(J))/PB2**3 - CR
C
      AT =  0.0
      BT = -COSD(T)* DELA(I)     /PA2**3  -  AT
      CT = -COSD(T)* YO          /PB1**3
      DT = -COSD(T)*(YO-DELB(I))/PB2**3  -  CT
C
C
      UX(J)=UX(J) + ZA(I)*AX + ZAP(I)*BX + ZB(I)*CX + ZBP(I)*DX
      UR(J)=UR(J) + ZA(I)*AR + ZAP(I)*BR + ZB(I)*CR + ZBP(I)*DR
   22 UT(J)=UT(J) + ZA(I)*AT + ZAP(I)*BT + ZB(I)*CT + ZBP(I)*DT
C
C
      BC(J)=UR(J)/(1.0+UX(J))
   21 CP(J)=-2.0*UX(J)-UX(J)**2-UR(J)**2-UT(J)**2
C
C
      THETA=T
      WRITE(6,25) XO, YO, THETA
   25 FORMAT(1H1,' BODY A     XO =',F5.3,'       YO =',F8.5,'
     -   THETA =',F6.2/
     -'   I          XA             RA             DRDXA          BC
     -UX             UR             CP             ZA             ZAP
     -UT'//)
      WRITE(6,333)(I,XA(I),RA(I),DRDXA(I),BC(I),UX(I),UR(I)
     - , CP(I), ZA(I), ZAP(I), UT(I), I=1,NS )
  333 FORMAT(1X,I2,10F13.5/)
      IF(T.LE.-90.0) GO TO 23
      T = T - 30.0
      GO TO 24
   23 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE GEOM(RAD,X,DX,NS,SLOPE)
C
C        GEOMETRY FOR M117 BOMB
C
      REAL L1,L2,L3,L4,L5
      INTEGER FINS
      DIMENSION RAD(31),X(31),SLOPE(31)
      L1=1.32288
      L2=2.54375
      L3=4.35000
      L4=4.94000
      L5=6.30
      RADT=0.262
      TAN1=0.7559289
      TAN2=0.1316500
      NSP1=NS+1
      X(1)=DX
      FINS=2
C
C        FINS=1  CORRESPONDS TO SIMULATED FIN BODY
C        FINS=2  CORRESPONDS TO STRAIGHT TAPER TAIL BODY
C
      DO 9 I=2,NSP1
      X(I)=X(I-1)+DX
C
C        NOSE SECTION
C
      IF(X(I-1).GT.L1) GO TO 1
      RAD(I-1)= SQRT(4.0-(X(I-1)-1.32)**2)-1.500
      SLOPE(I-1)=-(X(I-1)-1.32)/ SQRT(4.-(X(I-1)-1.32)**2)
      GO TO 9
C
C        MID-SECTION
C
    1 IF(X(I-1).GT.L2) GO TO 2
      RAD(I-1)=0.50
      SLOPE(I-1)=0.0
      GO TO 9
C
C        STRAIGHT REGION OF TAIL
C
    2 IF(X(I-1).GT.L3) GO TO 3
      RAD(I-1)=0.5-(X(I-1)-L2)*TAN2
      SLOPE(I-1)=-0.13165
      GO TO 9
    3 GO TO (4,7),FINS
C
C        SIMULATED FIN BODY
```

```
C
    4 IF(X(I-1).GT.L4) GO TO 5
      RAD(I-1)=RADT+(X(I-1)-L3)*(0.240/(L4-L3))
      SLOPE(I-1)=0.240/(L4-L3)
      GO TO 9
    5 IF(X(I-1).GT.L5) GO TO 6
      RAD(I-1)=0.502
      SLOPE(I-1)=0.0
      GO TO 9
    6 RAD(I-1)=0.0
      GO TO 9
C
C         STRAIGHT TAPER TAIL
C
    7 IF(X(I-1).GT.L5) GO TO 8
      RAD(I-1)=0.5-TAN2*(X(I-1)-L2)
      SLOPE(I-1)=-0.13165
      GO TO 9
    8 RAD(I-1)=0.0
      SLOPE(I-1)=0.0
C
    9 CONTINUE
C
      GO TO (10,12),FINS
   10 WRITE(6,11)
   11 FORMAT(1H1,'GEOMETRY FOR SIMULATED FIN BODY'//'  I     X(I)
     -RAD(I)      SLOPE(I)'/)
      GO TO 14
   12 WRITE(6,13)
   13 FORMAT(1H1,'GEOMETRY FOR STRAIGHT TAPER TAIL'//'  I    X(I)
     -RAD(I)      SLOPE(I)'/)
   14 WRITE(6,15) (I,X(I),RAD(I),SLOPE(I),I=1,NS)
   15 FORMAT(I5,3F10.5)
C
      RETURN
      END
```

```
      SUBROUTINE SIMQ(A,B,N,KS)
C
C     DOUBLE PRECISION SIMQ
C
C     DIMENSION A(1),B(1)
      DIMENSION A(1),B(1)
C
C    FORWARD SOLUTION
C
      TOL=0.0
      KS = 0
      JJ=-N
      DO 65 J=1,N
      JY=J+1
      JJ=JJ+N+1
      BIGA=0
      IT=JJ-J
      DO 30 I=J,N
C
C    SEARCH FOR MAXIMUM COEFFICIENT IN COLUMN
C
      IJ=IT+I
      IF( ABS(BIGA)- ABS(A(IJ))) 20,30,30
   20 BIGA=A(IJ)
      IMAX=I
   30 CONTINUE
C
C    TEST FOR PIVOT LESS THAN TOLERANCE (SINGULAR MATRIX)
C
      IF( ABS(BIGA)-TOL) 35,35,40
   35 KS = 1
      RETURN
C
C    INTERCHANGE ROWS IF NECESSARY
C
   40 I1=J+N*(J-2)
      IT=IMAX-J
      DO 50 NK=J,N
      I1=I1+N
      I2=I1+IT
      SAVE=A(I1)
      A(I1)=A(I2)
      A(I2)=SAVE
C
C    DIVIDE EQUATION BY LEADING COEFFICIENT
C
   50 A(I1)=A(I1)/BIGA
      SAVE=B(IMAX)
```

```
      B(IMAX)=B(J)
      B(J)=SAVE/BIGA
C
C    ELIMINATE NEXT VARIABLE
C
      IF(J-N)55,70,55
   55 IQS=N*(J-1)
      DO 65 IX=JY,N
      IXJ=IQS+IX
      IT=J-IX
      DO 60 JX=JY,N
      IXJX=N*(JX-1)+IX
      JJX=IXJX+IT
   60 A(IXJX)=A(IXJX)-(A(IXJ)*A(JJX))
   65 B(IX)=B(IX)-(B(J)*A(IXJ))
C
C    BACK SOLUTION
C
   70 NY=N-1
      IT=N*N
      DO 80 J=1,NY
      IA=IT-J
      IB=N-J
      IC=N
      DO 80 NK=1,J
      B(IB)=B(IB)-A(IA)*B(IC)
      IA=IA-N
   80 IC=IC-1
      RETURN
      END
```



(The reverse of this page is blank)

## REFERENCES


1. Smith, A. M. O., and Pierce, J.: Exact Solution of the Newmann Problem. Calculation of Non-Circulatory Plane and Axially Symmetric Flow about or within Arbitrary Boundaries. Douglas Aircraft Company. Report ES 26988, August, 1958.

2. Martin, Fred W.: Cross-Flow Corrected Axisymmetric Solution for Multiple Body Interference. AFATL-TR-71-109, August 1971.

3. Martin, Fred W.: Mutual Aerodynamic Interference Effects for Multiple Bodies by the Cross-Flow Corrections Method. AFATL-TR-71-69, June 1971.

4. Rankine, W. J. M.: "On the Mathematical Theory of Streamlines, Especially Those with Four Foci and Upwards." Philosophical Transactions, 1871.

5. Milne-Thomson, L. M.: Theoretical Hydrodynamics. The Macmillan Company, third edition, 1957.

6. Smith, Charles J.: "Mutual Aerodynamic Interference Effects for Two Similar Bodies," M.S. Thesis, Auburn University, Auburn, Alabama, December 1972.




(The reverse of this page is blank)

## INITIAL DISTRIBUTION

| | |
|---|---|
| DDC | 2 |
| AU (AUL/LSE-70-239) | 1 |
| ADTC (DLSOL) | 2 |
| TRADOC (ADTC-DO) | 1 |
| ASD (ENYS) | 1 |
| AFATL (DL) | 1 |
| HQ USAF (RDQRM) | 1 |
| HQ USAF (RDPA) | 1 |
| AFSC (SDWM) | 1 |
| AFDDL (FYS) | 1 |
| AFWL (SUL) | 1 |
| ASD (SD4E-2) | 1 |
| CINCUSAFE (LGWH) | 1 |
| OOAMA (MMNT) | 1 |
| OOAMA (MMEC) | 1 |
| AFMDC (TSL) | 1 |
| SMAMA (MMEM) | 1 |
| SAAMA (MMEAD) | 1 |
| NWC (Code 753) | 1 |
| NWC (Code 3051) | 1 |
| NWC (Code 406) | 1 |
| NMC (Code 5312) | 1 |
| NMC (Code 5331) | 1 |
| NSRDC (Code 5643) | 1 |
| NSRDC (Code 165) | 1 |
| NOL (Code 312) | 1 |
| NWL (Code KBB) | 1 |
| NASC (PMA-262-2) | 1 |
| ARMY/MAA (AMXSY-S) | 1 |
| ARMY/EACL (SMUEA-CL-PA) | 1 |
| SANDIA LABS (TECH LIB) | 1 |
| LTV AEROSPACE CORP | 1 |
| DR. MARTIN, AUBURN UNIV | 10 |
| BOEING CO. | 1 |
| AFATL/DLJ | 1 |
| AFATL/DLI | 1 |
| AFATL/DLW | 1 |
| AFATL/DLT | 1 |
| AFATL/DLM | 1 |
| AFATL/DLQ | 1 |
| AFATL/DLJA | 20 |
| AFATL/DLB | 1 |
| AFATL/DLK | 1 |
| AFATL/DLD | 1 |
| AFATL/DLR | 1 |
| Hq USAF/SAMI | 1 |



(The reverse of this page is blank)

(Security classification of title, body of abstract and indexing annotation must be entered when the overall report is classified)

1 ORIGINATING ACTIVITY (Corporate author)
Department of Aerospace Engineering
Auburn University
Auburn, Alabama

2a. REPORT SECURITY CLASSIFICATION
UNCLASSIFIED

2b. GROUP

3 REPORT TITLE
MUTUAL AERODYNAMIC INTERFERENCE EFFECTS FOR TWO AXISYMMETRIC BODIES

4 DESCRIPTIVE NOTES (Type of report and inclusive dates)
Final Report - September 1972 to June 1973

5 AUTHOR(S) (First name, middle initial, last name)
Fred W. Martin
Charles J. Smith
Grady Saunders

6 REPORT DATE
August 1973

7a. TOTAL NO. OF PAGES
52

7b. NO. OF REFS
7

8a. CONTRACT OR GRANT NO.
F08635-71-C-0090

b. PROJECT NO
2567

c. Task No.
02

d. Work Unit No.
015

9a. ORIGINATOR'S REPORT NUMBER(S)

9b. OTHER REPORT NO(S) (Any other numbers that may be assigned this report)
AFATL-TR-73-161

10 DISTRIBUTION STATEMENT
Distribution limited to U. S. Government agencies only; this report documents test and evaluation; distribution limitation applied August 1973 . Other requests for this document must be referred to the Air Force Armament Laboratory (DLJA), Eglin Air Force Base, Florida 32542.

11 SUPPLEMENTARY NOTE
Available in DDC

12. SPONSORING MILITARY ACTIVITY
Air Force Armament Laboratory
Air Force Systems Command
Eglin Air Force Base, Florida 32542

13 ABSTRACT
The mutual aerodynamic interference problem for two axisymmetric bodies has been analyzed using the image system technique. In order to facilitate this analysis, it has been assumed that small perturbation solutions are valid. It is further assumed that the external stores are slender bodies and that the mutual interference can be analyzed by first assuming a cross-flow solution. The image system in the cross-flow plane consists of source-sink pairs appropriately located by using the Milne-Thomson circle theorem. The actual three-dimensional source-sink pairs are displaced from the body axis according to the cross-flow image system. Their strengths are then determined by the Rankine method. Good agreement has been found between the theoretical and experimental results.

DD FORM 1 NOV 65 1473

UNCLASSIFIED
Security Classification

| 14 KEY WORDS | LINK A | | LINK B | | LINK C | |
|---|---|---|---|---|---|---|
| | ROLE | WT | ROLE | WT | ROLE | WT |
| Mutual Aerodynamic Interference | | | | | | |
| Mutual Aerodynamic Interference Effects | | | | | | |
| Axisymmetric Bodies | | | | | | |
| Image System Technique | | | | | | |
| Cross-Flow Solution | | | | | | |
| Milne-Thomson Circle Theorem | | | | | | |
| Three-Dimensional Source-Sink Pairs | | | | | | |
| Cross-Flow Image System | | | | | | |
| Rankine Method | | | | | | |
| Theoretical Results | | | | | | |
| Experimental Results | | | | | | |
| Theoretical/Experimental Results Comparison | | | | | | |